

Pseudo-Dynamic Approach to the Numerical Solution of Nonlinear Stationary Partial Differential Equations

Alexei Boulbitch

This article presents a numerical pseudo-dynamic approach to solve a nonlinear stationary partial differential equation (PDE) S with bifurcations by passing from S to a pseudo-time-dependent PDE T . The equation T is constructed so that the desired nontrivial solution of S represents a fixed point of T . The numeric solution of S is then obtained as the solution of T at a high enough value of the pseudo-time.


■ 1. Introduction: Soft Bifurcation of a Stationary Nonlinear PDE

The method described here can be applied to solve PDEs coming from different domains. However, it was initially developed to get the numerical solution of a stationary nonlinear PDE with a *bifurcation*. The method's application to a broader class of equations is briefly discussed at the end of the article.

The term “bifurcation” describes a phenomenon that occurs in some nonlinear equations that depend on one or several parameters. These equations can be algebraic, differential, integral or integro-differential. At some values of a parameter, such an equation may exhibit a fixed number of solutions. However, as soon as the parameter exceeds a critical value (referred to as the *bifurcation point*), the number of solutions changes and either new solutions emerge or some old ones disappear. To be specific, we discuss the case of dependence on a single parameter λ .

The new solutions can emerge continuously at the bifurcation point. The norm $\|\mathbf{u}\|$ of the solution $\mathbf{u}(\mathbf{x})$ exhibits a continuous though nonsmooth dependence on the parameter λ at the bifurcation point (left, Figure 1). An explicit example is in Section 4.5. A bifurcation at which the solution is continuous at the bifurcation point is referred to as *supercritical* or *soft*.

The behavior of the solution in the case of a *subcritical* or *hard* bifurcation is different: the norm of the solution is finite at the bifurcation point but has a jump discontinuity there (right, Figure 1).



▲ **Figure 1.** Soft versus hard bifurcation. In the case of the soft bifurcation, the solution $\mathbf{u}(\mathbf{x})$ has a continuous dependence of the solution norm $\|\mathbf{u}\|$ on the control parameter λ , with a kink at the bifurcation point, $\lambda = \alpha_0$. In contrast, in the case of a hard bifurcation, the solution is discontinuous at the bifurcation point.

In this article, we focus only on the case of a nonlinear PDE with soft bifurcations; some peculiarities of hard bifurcations are briefly discussed in Section 5.3.

In the most general form, a nonlinear PDE can be written as:

$$\mathbf{F}(\mathbf{u}_s(\mathbf{x}), \lambda) = 0. \quad (1)$$

Here $\mathbf{F} = (F_1, F_2, \dots, F_N)$ so that (1) indicates a system of N nonlinear PDEs; $\mathbf{u}_s = (u_{s_1}, u_{s_2}, \dots, u_{s_N})$ is an N -dimensional vector representing the dependent variable. The subscript s indicates that \mathbf{u}_s is the solution of a stationary equation. Further, \mathbf{x} is a D -dimensional vector. Finally, $\lambda \in \mathbb{R}$ is a real numerical parameter. The system of equations (1) is analyzed in a domain Ω subject to zero Dirichlet boundary conditions:

$$\mathbf{u}_s|_{\partial\Omega} = 0. \quad (2)$$

Also assume that

$$\mathbf{F}(0, \lambda) \equiv 0 \quad (3)$$

and thus $\mathbf{u}_{s0} \equiv 0$ represents a trivial solution of (1, 2).

It is convenient to separate out the linear part of the operator (1), which is often (though not always) representable in the form $\mathbf{L}[\mathbf{u}_s(\mathbf{x})] - \lambda \mathbf{u}_s(\mathbf{x})$ and to write it down in the following form:

$$\mathbf{L}[\mathbf{u}_s(\mathbf{x})] - \lambda \mathbf{u}_s(\mathbf{x}) + \mathbf{N}[\mathbf{u}_s(\mathbf{x})] = 0. \quad (4)$$

Here $\mathbf{L}[\mathbf{u}_s(\mathbf{x})]$ is a linear differential operator (such as, for example, the Laplace operator). Further, $\mathbf{N}[\mathbf{u}_s(\mathbf{x})]$ is the nonlinear part of the operator $\mathbf{F}[\mathbf{u}_s(\mathbf{x}), \lambda]$. The assumption that $\mathbf{u}_s \equiv 0$ solves equation (1) implies that $\mathbf{N}(0) \equiv 0$.

In its explicit form, we use the representation (4) only in Section 2.2, where we derive the critical slowing-down phenomenon. In all other cases, a general form of the dependence of equation (4) on λ is valid: $\mathbf{L} = \mathbf{L}[\mathbf{u}_s(\mathbf{x}), \lambda]$ and $\mathbf{N} = \mathbf{N}[\mathbf{u}_s(\mathbf{x}), \lambda]$. Nevertheless, we stick to the form (4) for simplicity, while the generalization is straightforward.

Let us also consider an auxiliary equation

$$\begin{cases} \mathbf{L}[\boldsymbol{\psi}_n(\mathbf{x})] = \alpha_n \boldsymbol{\psi}_n(\mathbf{x}) \\ \boldsymbol{\psi}_n|_{\partial\Omega} = 0 \end{cases} \quad (5)$$

that yields the linear part of the nonlinear equation (4). Equation (5) represents the eigenvalue problem, where the $\boldsymbol{\psi}_n$ are its eigenfunctions and the α_n are its eigenvalues, indexed by the discrete variable n , provided the discrete spectrum of (5) exists. Let us assume that at least a part of the spectrum of (5) is discrete. We assume here that n starts from zero: $n = 0, 1, 2, \dots$. The state with $n = 0$ is referred to as the *ground state*.

Without proofs, we recall a few facts from bifurcation theory [1] valid for soft bifurcations of such equations.

Assume that the trivial solution $\mathbf{u}_s = 0$ is stable for some values of λ . As soon as the parameter λ becomes equal to the smallest discrete eigenvalue α_0 of the auxiliary equation (5), this solution becomes unstable. As a result, a nontrivial solution $\mathbf{u}_s \neq 0$ branches off from the trivial one. In the close vicinity of the bifurcation point $\lambda = \alpha_0$, this solution has the asymptotics

$$\mathbf{u}_s = (\boldsymbol{\xi} \cdot \boldsymbol{\psi}_0) + O(\boldsymbol{\xi}^k), \quad (6)$$

where $\boldsymbol{\psi}_0 = (\psi_{01}(\mathbf{x}), \psi_{02}(\mathbf{x}), \dots, \psi_{0m}(\mathbf{x}))$ is the set of eigenfunctions of the equation (5) belonging to the eigenvalue α_0 . The vector $\boldsymbol{\xi} = (\xi_1, \xi_2, \dots, \xi_m)$ is the set of amplitudes. The scalar product $(\boldsymbol{\xi} \cdot \boldsymbol{\psi}_0)$ stands for the expression $(\sum_{i=1}^m \xi_i \psi_{0i}(\mathbf{x}))$. Here the index $i = 1, 2, \dots, m$ (where $m \leq N$) enumerates the eigenfunctions in the m -dimensional subspace of the functional space where (5) has a nonzero solution. The exponent k exceeds unity: $k > 1$.

There are a few methods available to determine $\boldsymbol{\xi}$. Listing them is out of the scope of this article. However, the simplest of these methods can be applied if there exists a generating functional $E\{\mathbf{u}\}$ enabling one to obtain the system of equations (1) as its minimum condition:

$$\frac{\delta E\{u\}}{\delta u_i} = 0; \quad (i = 1, 2, \dots, N), \quad (7)$$

where δ is the variational derivative. This functional we refer to as *energy* in analogy with physics. Substituting the representation (6) into the energy functional $E\{(\boldsymbol{\xi} \cdot \boldsymbol{\psi}_0)\}$ and integrating out the spatial coordinates, one finds the energy as a function of the amplitudes and parameter λ . Minimizing the energy with respect to the amplitudes ξ_i yields the system of m equations for the amplitudes, referred to as the *ramification equation*:

$$\frac{\delta E\{(\boldsymbol{\xi} \cdot \boldsymbol{\psi}_0) + \dots\}}{\delta \xi_i} = 0; \quad i = 1, 2, \dots, m. \quad (8)$$

Their solution is only accurate close to the bifurcation point $\lambda = \alpha_0$. Assuming that the bifurcation takes place with decreasing λ (as is the case in the following example), one finds the typical solution for the amplitudes,

$$\xi_i = \begin{cases} 0, & \lambda > \alpha_0 \\ \kappa_i(\alpha_0 - \lambda)^\beta, & \lambda \leq \alpha_0 \end{cases} \quad (9)$$

where κ_i and $\beta > 0$ are real numbers to be determined using the original equation. One of the methods to analytically find these parameters is discussed in Section 3. Further analytical methods may be found in [1]. This article focuses on finding these parameters numerically (Section 4.5).

All theorems and proofs for the preceding statements, along with more general methods of the derivation of the ramification equation, can be found in [1].

■ 2. Numerical Description of a Soft Bifurcation: A Problem and a Workaround

The bifurcation theory formulated so far is quite general: equation (1) can be differential, integral or integro-differential [1]. In what follows, we focus only on a more specific class of nonlinear partial differential equations.

The solution of the spectral system of equations (5) yields the bifurcation point $\lambda = \alpha_0$; the solutions (6) and (9) are only valid very close to this point. With increasing $|\lambda - \alpha_0|$, the solution soon deviates from the correct behavior quantitatively, and the solution $\mathbf{u}_s(\mathbf{x})$ often fails to resemble (6) even qualitatively. For this reason, to get the solution at some finite $|\lambda - \alpha_0|$ that would be correct both qualitatively and quantitatively, one needs to solve (1) numerically.

In the case of a hard bifurcation, none of the machinery of the theory of soft bifurcations described so far works. Studying the bifurcation numerically often becomes the only possibility.

However, the direct numerical solution of nonlinear equations like (1) and (4) with some nonlinear solvers only returns the trivial solution $\mathbf{u}_s(\mathbf{x}) \equiv 0$ for equation (4), even at the values of the parameter λ at which the trivial solution is unstable and a stable nontrivial solution already exists.

A plausible reason may be as follows: the solver starts to construct the PDE solution from the boundary. Here, however, the boundary condition $\mathbf{u}_s|_{\partial\Omega} = 0$ is already part of the trivial solution. Thus the solver appears to be placed at the true solution of the equation and is then unable to climb down from it.

To find a nontrivial solution, one needs to use a method that would start from some initial approximation that, even if rough, should be quite different from the trivial solution. Furthermore, this method should converge to the nontrivial solution by a chain of successive steps.

□ 2.1. A Pseudo-Dynamic Equation

One can do this with the pseudo-dynamic approach formulated in the present article.

Let us introduce pseudo-time t . The word "pseudo" indicates that t is not real time. It just represents a technical trick that helps with the simulation. Assume now that the dependent variable is a function of both the set of spatial coordinates \mathbf{x} and the pseudo-time: $\mathbf{u}(\mathbf{x}, t)$. Instead of the stationary equation (1), let us study the behavior of the pseudo-time-dependent equation:

$$\begin{cases} \frac{\partial \mathbf{u}(\mathbf{x}, t)}{\partial t} = \mathbf{F}(\mathbf{u}(\mathbf{x}, t), \lambda) \\ \mathbf{u}(\mathbf{x}, t) |_{x \in \partial \Omega} = 0 \\ \mathbf{u}(\mathbf{x}, 0) = \mathbf{u}_0(\mathbf{x}) \neq 0 \end{cases} \quad (10)$$

One solves equation (10) with a suitable nonzero initial condition $\mathbf{u}_0(\mathbf{x})$. Let us stress that the solution $\mathbf{u}(\mathbf{x}, t)$ of the time-dependent equations (10) is not the same as the solution $\mathbf{u}_s(\mathbf{x})$ of the stationary equation (1).

One could also construct the pseudo-time-dependent equation as follows: $\partial \mathbf{u} / \partial t = -\mathbf{F}$, that is, with a minus sign in front of \mathbf{F} . The idea of such an extension is that either $\partial \mathbf{u} / \partial t = \mathbf{F}$ or $\partial \mathbf{u} / \partial t = -\mathbf{F}$ exhibits a fixed point, so that $\mathbf{u}(\mathbf{x}, t) \rightarrow \mathbf{u}_s(\mathbf{x})$, while the other diverges as $t \rightarrow \infty$. By trial and error, one chooses the equation whose solution converges to the fixed point $\mathbf{u}_s(\mathbf{x})$.

The operator $\mathbf{F}(\mathbf{u}(\mathbf{x}, t), \lambda)$ has not yet been specified; for definiteness let us assume that the fixed point $\mathbf{u}_s(\mathbf{x})$ at $t \rightarrow \infty$ takes place for equation (10), that is, with the plus sign in front of \mathbf{F} .


The convergence of the solution of the dynamic equation to the fixed point $\mathbf{u}_s(\mathbf{x})$ enables one to apply the following strategy. Instead of the static equation (1), which is difficult to solve numerically, one simulates the quasi-dynamic equation (10) using a suitable time-stepping algorithm.

The advantage of this approach is in the possibility of starting the simulation from an arbitrary distribution $\mathbf{u}_0(\mathbf{x}, 0) \neq 0$ chosen as the initial condition, provided it agrees with the boundary conditions. From the very beginning, such a choice takes one away from the trivial solution. The time-stepping process takes the initial condition for each step from the previous solution. The solution starting from any function $\mathbf{u}_0(\mathbf{x}, 0) \neq 0$ gradually converges to $\mathbf{u}_s(\mathbf{x})$ with time if $\mathbf{u}_0(\mathbf{x}, 0)$ belongs to its attraction basin.

After having obtained the solution of the pseudo-time-dependent equation, one approximates the function $\mathbf{u}_s(\mathbf{x})$, as $\mathbf{u}_s(\mathbf{x}) \approx \mathbf{u}(\mathbf{x}, t_{\max})$ at a large enough value of the pseudo-time $t = t_{\max}$. The meaning of the words "large enough" is clarified in Section 4.3.

The approach can be given a pictorial interpretation (Figure 2). In the infinite-dimensional functional space, let $\{\psi_0, \psi_1, \psi_2, \dots\}$ be an infinite set of basis functions. Then the function $\mathbf{u}(\mathbf{x}, t)$ can be represented as

$$\mathbf{u}(\mathbf{x}, t) = \xi_0(t) \psi_0(\mathbf{x}) + \xi_1(t) \psi_1(\mathbf{x}) + \xi_2(t) \psi_2(\mathbf{x}) + \dots \quad (11)$$




▲ **Figure 2.** Schematic view of the 3D projection of the infinite-dimensional functional space with a trajectory from the initial state (blue dot) to the fixed point (red dot).

The trajectory $(\xi_0(t), \xi_1(t), \xi_2(t), \dots)$ in this space goes from the initial state $(\xi_0(0), \xi_1(0), \xi_2(0), \dots)$ to the final state $(\xi_0(\infty), \xi_1(\infty), \xi_2(\infty), \dots)$, as shown by the two dots.

The time derivative $\partial \mathbf{u} / \partial t$ represents the velocity of the motion of a point through this space, while \mathbf{F} can be regarded as a force driving this point. Thus equation (10) can be interpreted as describing a driven motion of a massless point particle with viscous friction through the functional space. In these terms, the condition (1) means that the driving force is equal to zero at some point of the space, which is the location of the fixed point of the nonlinear equation (10).

If the energy functional $E\{\mathbf{u}\}$ for equation (1) exists, one can make one further step in the interpretation (Figure 3).



▲ **Figure 3.** Schematic view of the energy functional E as the function of the coordinate in the functional space (A) above and (B) below the bifurcation point. The cross section of the infinite-dimensional space along a single coordinate ξ is shown. The points show initial positions of the particle, while the arrows indicate its motion to the nearest minimum of the potential well.

Indeed, according to the definition given, equation (1) delivers a minimum to the energy functional. In this case, one can regard the dynamic equation (10) as describing a viscous motion of the massless point particle along a hypersurface in the $1 + \infty$ -dimensional space, (E, ξ_0, ξ_1, \dots) , the surface forming a potential well. The motion goes from some initial position to the minimum of the potential well as shown schematically in Figure 3. Above the bifurcation, this minimum only corresponds to the trivial solution (A) situated at $\xi = 0$. Below the bifurcation, the energy hypersurface exhibits a new configuration with new minima, while the previous minima vanish. As a result, below the bifurcation, the point particle moves from the initial position (shown by dots in Figure 3) to one of the newly formed minima (as the red and green arrows show in B). The functional space has infinite dimension, and essential features of the numeric process may involve several dimensions. The $1 + 1D$ representation displayed in Figure 3 is therefore oversimplified and only partially represents the bifurcation phenomenon.

Equation (10) can be rewritten as:

$$\begin{cases} \frac{\partial \mathbf{u}(\mathbf{x}, t)}{\partial t} = \mathbf{L}[\mathbf{u}(\mathbf{x}, t)] - \lambda \mathbf{u}(\mathbf{x}, t) + \mathbf{N}[\mathbf{u}(\mathbf{x}, t)] \\ \mathbf{u}(\mathbf{x}, t) |_{\mathbf{x} \in \partial \Omega} = 0 \\ \mathbf{u}(\mathbf{x}, 0) = \mathbf{u}_0(\mathbf{x}) \neq 0 \end{cases} \quad (12)$$

Though lacking a stationary nonlinear solver at present, Mathematica offers the option `MethodOfLines`, efficiently applicable to dynamic equations like (12). This method is applied everywhere in the rest of this article.

The evident penalty of this approach is that the computation time can become large, especially in the vicinity of the bifurcation point; this peculiarity is discussed next.

□ 2.2. A Critical Slowing Down

Close to the critical point $\lambda = \alpha_0$, the relaxation of the solution to the fixed point $\mathbf{u}_s(\mathbf{x})$ dramatically slows down. This is referred to as *critical slowing down*. Its origin is illustrated in Section 4. To simplify the argument, let us consider a single equation $F(u(\mathbf{x}), \lambda) = 0$ with the one-component dependent variable $u(\mathbf{x}, t)$ that still depends on the D -dimensional coordinate \mathbf{x} . The generalization for a system of equations is straightforward, though a bit cumbersome.

According to (6), close to the bifurcation point, one can look for the solution of equation (12) in the form:

$$u(\mathbf{x}, t) = \xi(t) \psi_0(\mathbf{x}). \quad (13)$$

Ignore the higher-order terms, assuming that ξ is small. Substitute (13) into the first equation (12) and linearize it. Here one should distinguish between the case at $\lambda > \alpha_0$, where the linearization should be done around $\xi = 0$, and that at $\lambda \leq \alpha_0$, where one linearizes with the center at $\xi \sim (\alpha_0 - \lambda)^\beta$ (the second line of equation 9). In the former case, one finds

$$\psi_0(\mathbf{x}) \frac{\partial \xi(t)}{\partial t} = \xi(t) \{ \mathbf{L}[\psi_0(\mathbf{x})] - \lambda \psi_0(\mathbf{x}) \}.$$

Making use of (5), one finally obtains the dynamic equation for $\xi = \xi(t)$ at $\lambda > \alpha_0$:

$$\frac{\partial \xi(t)}{\partial t} = -(\lambda - \alpha_0) \xi(t), \quad (14)$$

implying that $\xi(t) \sim \exp(-t/\tau)$, and the relaxation time τ has the form $\tau = (\lambda - \alpha_0)^{-1}$.

At $\lambda \leq \alpha_0$, analogous but somewhat more lengthy arguments give the characteristic time, twice as small as that above the critical point. One comes to the relation:

$$\tau = \frac{1}{|\alpha_0 - \lambda|} \begin{cases} 1, & \lambda > \alpha_0 \\ \frac{1}{2}, & \lambda \leq \alpha_0 \end{cases} \quad (15)$$

One can see that the relaxation time τ diverges with $\lambda \rightarrow \alpha_0$ from both sides. From the practical point of view, this suggests increasing the simulation time according to (15) near the critical point.

The result (15) is valid for equation (12), in which the linear part of the pseudo-dynamic equation has the form $\mathbf{L}[\mathbf{u}(\mathbf{x}, t)] - \lambda \mathbf{u}(\mathbf{x}, t)$. That is, the parameter λ enters this equation only linearly, in the form of the product $\lambda \mathbf{u}(\mathbf{x}, t)$. In the general case $\mathbf{L} = \mathbf{L}[\mathbf{u}(\mathbf{x}, t), \lambda]$, one still finds diverging relaxation time $\tau \sim |\alpha_0 - \lambda|^{-1}$, though the factors (such as 1 above, and 1/2 below the bifurcation point) may be different.

The phenomenon of critical slowing down was first discussed in the framework of the kinetics of phase transitions [2].

■ 3. Example: A 1D Ginzburg–Landau Equation

As an example, let us study the 1D PDE:

$$\begin{cases} \frac{\partial^2 u(x)}{\partial x^2} = 2[\lambda + U(x)] u(x) + u^3(x) \\ u(\pm\infty) = 0 \end{cases} \quad (16)$$

where $u = u(x)$ is the dependent variable of the single coordinate x . This equation exhibits a cubic nonlinearity $u^3(x)$. A classical Ginzburg–Landau equation only has constant coefficients for the terms $u(x)$ and $u^3(x)$. In contrast, equation (16) possesses the inhomogeneity $U(x)$ with


$$U(x) = -\frac{1}{\cosh^2(x)}, \quad (17)$$

shown by the solid line in Figure 4. It thus represents a nonhomogeneous version of the Ginzburg–Landau equation. One can see that (16) has the trivial solution $u(x) \equiv 0$.


```

Plot[{-2 Cosh[x]^-2, Cosh[x]^-1}, {x, -10, 10},
  PlotTheme -> "Classic", PlotRange -> {-2.1, 1.3},
  PlotStyle -> {Red, {Dashed, Blue}},
  AxesLabel -> {Style["x", Italic], Row[{U[x], " ", " ", ψ₀[x]}]},
  AxesStyle -> Arrowheads[0.03],
  PlotLegends ->
    Placed[LineLegend[{U[x], ψ₀[x]}, LegendFunction -> "Frame",
      Scaled[{0.8, 0.85}]]
    ], ImageSize -> 350]

```



▲ **Figure 4.** The potential $U(x)$ from equation (17) (solid, red) and the solution $\psi_0(x)$ of the auxiliary equation (18) (dashed, blue).

Equations (16) and (17) play an important role in the theory of the transformation of types of domain walls into one another [3].

The auxiliary equation (5) in this case takes the following form:

$$\begin{cases} \frac{\partial^2 \psi_n(x)}{\partial x^2} = 2 \{ \alpha_n + U(x) \} \psi_n(x) \\ \psi_n(\pm\infty) = 0 \end{cases} \quad (18)$$

where $n = 0, 1, 2, \dots$ enumerates the eigenvalues and eigenfunctions belonging to the discrete spectrum. One can see that equation (18) represents the Schrödinger equation $u''(x) + [E + U(x)] u(x) = 0$ [4] with potential well $U(x)$ (17) and energy $E = -2\alpha_n$.

The exact solution of the auxiliary equation (18) is known [3, 4]. It has two discrete eigenvalues when $n = 0$ and $n = 1$, and the ground-state ($n = 0$) solution has the form

$$\alpha_0 = \frac{1}{2}; \quad \psi_0(x) = \frac{1}{\cosh(x)}, \quad (19)$$

which can be easily checked by direct substitution.

The energy functional generating the Ginzburg–Landau equation (16, 17) has the form:

$$E\{u\} = \int_{-\infty}^{\infty} \left\{ \frac{1}{2} \left(\frac{\partial u_s(x)}{\partial x} \right)^2 + [\lambda - \cosh^{-2}(x)] u_s^2(x) + \frac{1}{4} u_s^4(x) \right\} dx. \quad (20)$$

Equation (6) can be written as $u_s(x) \approx \xi \psi_0(x)$. Substituting that into equation (20) for the energy, eliminating the term with the derivative using equation (18) and applying the Gauss theorem, one finds the energy as a function of the amplitude ξ :

$$E(\xi) = 2 \left(\lambda - \frac{1}{2} \right) \xi^2 + \frac{1}{3} \xi^4. \quad (21)$$

The *ramification* equation takes the form $\partial E / \partial \xi = 0$:

$$\left(\lambda - \frac{1}{2} \right) \xi + \frac{1}{3} \xi^3 = 0, \quad (22)$$

with the following solution for the amplitude:

$$\xi(\lambda) = \begin{cases} 0, & \lambda > 1/2 \\ \pm \sqrt{3} \left(\frac{1}{2} - \lambda \right)^{1/2}, & \lambda \leq 1/2 \end{cases} \quad (23)$$

■ 4. Numerical Solution of the Ginzburg–Landau Equation

□ 4.1. Pseudo-Time-Dependent Equation

Let us now look for the numerical solution of equation (16). The problem to be solved is to find the point of bifurcation and the overcritical solution at $\lambda < 1/2$. The pseudo-time-dependent equation can be written as:

$$\begin{cases} \frac{\partial u(x,t)}{\partial t} = \frac{\partial^2 u(x,t)}{\partial x^2} - 2 \left\{ \lambda - \frac{1}{\cosh^2(x)} \right\} u(x,t) - u^3(x,t) \\ u(\pm \infty, t) = 0 \end{cases} \quad (24)$$

The choice of the initial condition is not critical, provided it is nonzero. The method of lines employed in the following is relatively insensitive to whether or not the initial condition precisely matches the boundary conditions. We demonstrate its solution with three initial conditions

$$\begin{aligned} u(x, 0) &= 0.2 \exp(-x^2), \\ u(x, 0) &= 0.2 \exp[-(x-3)^2], \\ u(x, 0) &= 0.3 \times 10^{-4} (x+10)(10-x)^3, \end{aligned}$$

in the `Manipulate` in the next section.

□ 4.2. Solution within a Finite Domain

The method of lines is applied here since it can solve nonlinear PDEs, provided these equations are dynamic, which is exactly the case within the pseudo-time-dependent approach.

To address the problem numerically, let us start with the initial conditions taken at a finite distance, rather than at infinity. The distance must be greater than the characteristic dimension of the equation, which is the distance for which $u(x)$ exhibits a considerable variation. For the Ginzburg–Landau equation (16), the characteristic dimension is defined by the width of the potential $U(x)$ for (17), which is about 1. That is, let us start with the boundaries at $x = \pm L$ with $L = 10$. We check the quality of the result obtained with such a boundary later.

To obtain a precise enough solution, one needs to make a spatial discretization providing a step comparable to the characteristic dimension of the equation, which we just saw is of the order of 1. Therefore, a step that is small enough can be a few times L . The value $5L$ appears to be enough.

The following code solves the equation. To keep the discretization with the step comparable to the characteristic equation dimension, we chose "MinPoints" $\rightarrow 5L$.

To avoid conflicts with variables that may have been previously set, this notebook has the setting Evaluation ► Notebook's Default Context ► Unique to This Notebook.

```
equation[λ_] :=
  D[u[x, τ], τ] == D[u[x, τ], {x, 2}] - 2  $\left( \lambda - \frac{1}{\text{Cosh}[x]^2} \right)$  u[x, τ] -
  u[x, τ]^3

solution[λ_, τm_, L_, initialCondition_] :=
  NDSolve[{equation[λ], u[-L, τ] == 0, u[L, τ] == 0,
    initialCondition}, u[x, τ], {x, -L, L}, {τ, 0, τm},
  Method → {"MethodOfLines",
    "SpatialDiscretization" →
    {"TensorProductGrid", "MinPoints" → 5L}}
```

According to Section 2, the time-dependent solution $u(x, t)$ obtained converges to the solution $u_s(x)$ of the stationary problem $u_s(x) = u(x, t \rightarrow \infty)$. In practice, however, one can instead take some finite $t = t_{\max}$ value, provided that it is large enough.

We solve the pseudo-dynamic equation (24) with each of the three initial conditions stated before.

Further, in order to give the feeling of the method, we visualize and animate the solution, varying λ as well as the initial conditions. This requires a few comments. As discussed in Section 2.2, the maximum time of simulation t_{\max} strongly depends on λ . This is accounted for by introducing $t_{\max}(\lambda) = t_0 |0.5 - \lambda|^{-1}$ according to (15), where $t_0 = 3.2$ was chosen by trial so that the simulation does not last too long, but also so that the value of $t_{\max}(\lambda)$ always ensures the convergence for any combination of λ and initial condition.

In the simulations, you can observe two essential features of the present method.

First, near the fixed point, the solution converges more slowly and the curve gradually appears to stop changing.


Second, near the critical point, close to $\lambda \approx 0.5$, the critical slowing down (see Section 2.2) takes place, which requires considerably longer to approach the fixed point. In the animation, the curve evolves much more slowly at $\lambda = 0.4$ and $\lambda = 0.6$, and the convergence, therefore, requires much more time.

In the `Manipulate`, choose one of the three initial conditions and a value of λ . Click the button with the arrow to start the animation. The value of the current time is shown at the top-left corner. The distribution shown by the blue curve at $t = 0$ corresponds to the initial condition, while at $t > 0$ the animation shows its further evolution.

```

Manipulate[
  ic = Switch[icNumber,
    1, u[x, 0] == 0.2 Exp[-x^2],
    2, u[x, 0] == 0.2 Exp[-(x - 5)^2],
    3, u[x, 0] == 0.3 * 10^-4 (x + 10) (10 - x)^3
  ];
  tmax = 3.2 / Abs[0.5 - λ];
  Plot[
    Evaluate[u[x, τ] /. solution[λ, tmax, 10, ic] /. τ → tmax t],
    {x, -10, 10}, PlotTheme → "Classic",
    PlotRange → {{-10.2, 11}, {0, 1.}},
    PlotStyle → {Blue, Thick},
    AxesLabel → {Style[x, 16],
      Style[u[x, Style["t", Italic]], 16]},
    TicksStyle → Directive[12], AxesStyle → Arrowheads[0.03],
    ImageSize → 400],
  Row[{
    Control[{{icNumber, 1, "initial condition number"},
      {1, 2, 3}}],
    Spacer[50],
    Control[{{λ, 0.2, "λ value"}, {0.2, 0.3, 0.4, 0.6, 0.7}}]
  ]],
  {{t, 0,
    Dynamic@Panel[
      Style[Row[{Style["t", Italic], " = ",
        NumberForm[tmax t, {3, 2}]}], 12, Blue],
      ImageSize → {70, 40}}, 0, 1, 0.01, ImageSize → {300, 15},
    ControlType → Animator, AnimationRepetitions → 1,
    AnimationRunning → False},
  SaveDefinitions → True
]

```



For each of the three initial conditions, the solution converges to the same bell-shaped curve. One can make sure that for low λ , the solution $u(x)$ is nonzero. However, for λ greater than about 0.5, the solution is trivial.

□ 4.3. The Solution Norm and the Convergence Control

To get an accurate solution, one needs to control the convergence as the pseudo-time increases. Here we control the convergence by analyzing the behavior of the integral

$$\|u(t)\| = \left(\frac{1}{2L} \int_{-L}^L u^2(x, t) dx \right)^{1/2} \quad (25)$$

(the norm of the solution in Hilbert space) at a fixed value of the parameter λ as a function of t_{\max} . The norm $\|u\|$ is zero above the bifurcation but nonzero below it.


We show how $\|u(t_{\max})\|$ depends on the time limit $t = t_{\max}$ at three fixed values of the control parameter λ : 0.43, 0.48 and 0.495, which are all below the bifurcation point $\lambda = 0.5$.

The following code makes a nested list containing three sublists corresponding to the three λ values. Each sublist consists of pairs $\{t_{\max}, \|u(t_{\max})\|\}$ at different values of the simulation time t_{\max} , which increases from 10 to approximately 3000. The exponential rate of increase is chosen so as to make the plot on a semilogarithmic scale look equally spaced (Figure 5).

```

ListPlot[
  Table[
    {tmax,
      (NIntegrate[
        u[x,  $\tau$ ]2 /. solution[#, tmax, 10,
          u[x, 0] == 0.2 Exp[-x2]] /.  $\tau$  → tmax,
        {x, -10, 10}] / 20)1/2,
      {tmax, 10^Range[1., 3.5, 1/12]}
    ] /. {x_, {y_}} → {x, y} & /@ {0.43, 0.48, 0.498},
  PlotStyle → {Blue, Red, Brown}, PlotMarkers → Automatic,
  PlotTheme → "Scientific",
  ScalingFunctions → {"Log", Automatic}, Axes → None,
  FrameLabel → {Style["tmax", 12],
    Style[Row[{"||", Style["u", Italic], "||"}, 12, Black]},
  FrameTicksStyle → Directive[{Black, 12}],
  PlotLegends → Placed[PointLegend[{
    Style["0.430", 12],
    Style["0.480", 12],
    Style["0.498", 12]}, LegendFunction → "Frame",
    LegendLabel → Style[" $\lambda$ ", 12]], Right]
]

```



▲ **Figure 5.** Semilogarithmic plots of the Hilbert norm $\|u\|$ of the solution for $\lambda = 0.43$ (disks), $\lambda = 0.48$ (squares) and $\lambda = 0.498$ (diamonds) depending on the simulation time, t_{\max} .

There is convergence for all three values of λ . However, the value of t_{\max} for which the convergence is satisfactory depends on λ . For example, at $\lambda = 0.48$ the solution at t_{\max} slightly exceeding 100 is already near convergence. Thus, with $t_{\max} = 200$, one can be sure that the solution is satisfactory. We use this in Section 4.4 to determine the expression for t_{\max} accounting for the critical slowing down.

In contrast, the solution for $\lambda \approx 0.498$ shows some evolution even at $t_{\max} = 1000$.

□ 4.4. The Critical Slowing Down in the Numeric Process

As we showed in Section 2.2, the t_{\max} value that gives satisfactory convergence depends on $|\lambda - \alpha_0|$. To get an accurate solution, t_{\max} must considerably exceed the relaxation time τ . For example, in the calculation of the result shown in Figure 4, substituting $\lambda = 0.48$ and $\alpha_0 \approx 0.5$ into (15), one finds $\tau \approx 25$, while the convergence only becomes good enough at $t_{\max} \approx 200$, which is eight times greater than τ . This implies that to find an accurate solution in the close vicinity of the bifurcation point, one has to define t_{\max} depending on λ by

$$t_{\max} = \frac{8}{[(\alpha_0 - \lambda)^2 + \epsilon]^{1/2}}, \quad (26)$$

where $\epsilon = 10^{-6}$ is the regularization parameter.

□ 4.5. In Search of the Bifurcation Point

The bifurcation point can be found by analyzing the same integral calculated at $t = t_{\max}$ in (26). Let us denote $\|u(t_{\max})\| = \|u\|$. This time we study the integral $\|u\|$ as a function of the parameter λ .

The transition from $u_s \equiv 0$ to $u_s(x) \neq 0$ occurs at the bifurcation point. Accordingly, the integral $\|u\|$ at this point changes from $\|u\| \equiv 0$ to $\|u\| \neq 0$.

To find the critical point, bifurcation theory (23) predicts the norm $\|u\|$ to be expressed in the form:

$$\|u\| = \begin{cases} 0, & \lambda > \alpha_0 \\ \eta_0 (\alpha_0 - \lambda)^{0.5}, & \lambda \leq \alpha_0 \end{cases} \quad (27)$$

We find the constant parameters η_0 and α_0 by fitting.


We now find the numerical solution of the equation (16) as a function of the control parameter λ ; the norm $\|u\|$ obtained from this solution depends on λ . We vary λ from 0.45 to 0.52 to create a list consisting of pairs $\{\lambda, \|u\|\}$. The most critical region for $\|u\|$ dependence is close to the critical point, so the points there are taken to be about 10 times more dense. This list is fitted to the function (27). The list is plotted with the analytic function obtained by fitting (Figure 6).

```

tmaxfn[λ_, ε_] := 8 / ((0.5 - λ)^2 + ε)^1/2

Module[
  {ε = 10^-6, listNorm2, fit},
  listNorm2 =
  Table[
    {λ,
     (NIntegrate[
       u[x, τ]^2 /. solution[λ, tmaxfn[λ, ε], 10,
        u[x, 0] == 0.2 Exp[-x^2]] /. τ → tmaxfn[λ, ε],
       {x, -10, 10}] / 20)^1/2}, {λ, 0.45, 0.52, 0.001}] /.
  {x_, {y_}} → {x, y};
  Column[{
    fit = FindFit[listNorm2, If[λ >= α0, 0, η0 ((α0 - λ)^2 + ε)^1/4],
      {η0, {α0, 0.49}}, λ],
    "",
    Show[{
      ListPlot[listNorm2, PlotStyle → Red,
        PlotTheme → "Scientific", PlotRange → {0, 0.14},
        Axes → None, FrameTicksStyle → Directive[{Black, 10}],
        FrameLabel → {Style["λ", 16],
          Row[{"|", Style["u", 16, Italic], "|"}]},
        PlotLegends →
        Placed[PointLegend[
          {Style["numeric solution", 12,
            FontFamily → "Times"]}], Scaled[{0.77, 0.77}]]
      ],
      Plot[If[λ > α0, 0, η0 (α0 - λ)^0.5] /. fit, {λ, 0.46, 0.54},
        PlotStyle → {Blue}, PlotLegends → Placed[LineLegend[{
          Style[Row[{" fit with α0 ≈ ",
            NumberForm[fit[[2, 2]], {5, 4}]]], 12,
          FontFamily → "Times"
        }], Scaled[{0.75, 0.6}]]]
    }, ImageSize → 400]
  ]
]

```

▲ **Figure 6.** Behavior of the Hilbert norm $\|u\|$ of the solution in the vicinity of the bifurcation point. Dots show the integrals (25), while the solid line indicates the result of fitting with the relation (27), yielding $\alpha_0 \approx 0.4997$.

The values of the integrals $\|u\|$ at various λ are shown by the red dots in Figure 6, while its fitting curve is shown by the solid blue curve. The fitted value of the bifurcation point is $\alpha_0 \approx 0.4997 \approx 0.50$ and $\eta_0 \approx 0.56$.

We used equation (26) for the t_{\max} used in the solution. However, this equation depends on the spectral value α_0 . In the present case, the value $\alpha_0 = 1/2$ was known, which considerably simplifies the task. In general, the value of α_0 is only established in the course of the fitting procedure, requiring an iterative approach. For the first simulation, we fix some large enough value of t_{\max} independent of λ and obtain a fit. This fit gives the first guess for α_0 , which can then be used for the simulation with the equation (26). This procedure can be repeated until a satisfactory α_0 is achieved.

□ 4.6. Varying the Boundary


To check how the choice of the boundary affects the results, we solve the problem by gradually increasing L (Figure 7). (This takes some time.)

```

fitRes[L_] := Module[
  { $\epsilon = 10^{-6}$ },
  FindFit[
    Table[
      { $\lambda$ ,
        (NIntegrate[u[x,  $\tau$ ]2 /. solution[ $\lambda$ , tmaxfn[ $\lambda$ ,  $\epsilon$ ],
          L, u[x, 0] = 0.2 Exp[-x2]] /.
           $\tau \rightarrow$  tmaxfn[ $\lambda$ ,  $\epsilon$ ], {x, -L, L}] / (2 L)1/2},
      { $\lambda$ , 0.45, 0.52, 0.001}] /. {x_, {y_}}  $\rightarrow$  {x, y},
    If[ $\lambda \geq \alpha_0$ , 0,  $\eta_0 ((\alpha_0 - \lambda)^2 + \epsilon)^{1/4}$ ], {
       $\eta_0$ , { $\alpha_0$ , 0.49}},  $\lambda$ ][[2, 2]]
  ] // Quiet

ListLogLogPlot[
  Table[{L, fitRes[L]}, {L, {5, 10, 20, 50, 100, 300, 1000}}] /.
  {x_, y_}  $\rightarrow$  {x, Abs[y - 0.5] 104}, PlotTheme  $\rightarrow$  "Classic",
  PlotStyle  $\rightarrow$  {PointSize[0.015], Blue}, Frame  $\rightarrow$  True,
  FrameLabel  $\rightarrow$  {Style["L", 12, Italic],
  Style[Row[{Abs[ $\alpha_0$  - 0.5], "  $\times$  ", "10-4"}], 12]}]

```



▲ **Figure 7.** A double-logarithmic plot showing the convergence of the bifurcation point with increasing L .

Figure 7 displays the error in the spectral value obtained by the numerical process. As one could have expected, with the increase of L , it decreases from 5×10^{-4} to about 1×10^{-4} .

■ 5. Discussion

The preceding example has shown the application of the pseudo-dynamic approach for solving a 1D nonlinear PDE with zero boundary conditions that exhibits a supercritical (soft) bifurcation. That simple problem was chosen to keep the processing time as short as possible. Now possible extensions are discussed.

□ 5.1. Nonzero Boundary Conditions

Recall that zero boundary conditions often (if not always) represent a problem for a nonlinear solver. Starting from $u = 0$ along the boundary, such a solver often only returns the trivial solution, since zero is, indeed, the solution of the equation considered here. For this reason, a solution to a problem like the one discussed in this article necessarily requires some specific approach that can converge to a nontrivial solution. It is for this type of equation that the approach presented here has been developed.

One should, however, make two comments.

First, there are numerous problems where the bifurcation takes place from a solution $u_{s0}(x)$ that is nonzero. The boundary condition in this case has the form $u|_{\partial\Omega} \neq 0$. A trivial observation shows that one comes back to the original problem by the shift $\delta u_{s0}(x) = u_s(x) - u_{s0}(x)$.

Second, the approach formulated here can be applied to nonlinear equations with no bifurcation. These equations can have boundary conditions that are either zero or nonzero. Indeed, such equations can often be solved by a nonlinear solver if one is available. Among other approaches, the present one can be applied; the nonzero boundary conditions are not an obstacle for the transition to the pseudo-time-dependent equation.

Though the present approach takes longer, in certain cases it is preferable; for example, when due to a strong nonlinearity the nonlinear solvers fail. The solver moves along the pseudo-time parameter in small steps δt from $\mathbf{u}(x, t)$ to $\mathbf{u}(x, t + \delta t)$, gradually passing from the initial condition to the final solution. Such a slow ramping can be stable.


□ 5.2. Dimensionality

The space dimensionality does not limit the application of our approach (for 2D examples, see [5, 6]).

□ 5.3. A Supercritical (Soft) versus a Subcritical (Hard) Bifurcation

In the case of a soft bifurcation, the energy can have only one type of minimum, as shown in Figure 2 describing the convergence either to the trivial or the nontrivial solution. The trajectory always flows into the minimum along the steepest slope of E . The minimum is a fixed point.

An essentially different situation occurs for a hard bifurcation, when the hypersurface $E(\xi_1, \xi_2, \dots)$ may have multiple minima. Figure 8 (A) shows a schematic cross section of the infinite-dimensional functional along the (ξ, E) plane, leaving out all other dimensions. This cross section shows the situation with minima of different types. One of these minima is more pronounced than the others. The arrows schematically indicate the trajectories in the functional space. These start from the initial conditions displayed by the dots in Figure 8 (A, B) and converge to the minima (Figure 8 A). The green arrow shows the convergence of the process to the principal minimum, while the red one converges to a secondary minimum.



▲ **Figure 8.** Schematic view of the energy functional along a direction ξ of the functional space, where it exhibits a metastable minimum (A). The green point schematically indicates the initial condition starting from which the solution converges to the one corresponding to the principal energy minimum (green arrow), while the red dot shows the initial condition leading to the convergence to the secondary minimum. (B) The trajectory ends at an inflection point.

As a result, depending on the choice of initial condition, some solution trajectories may end up at a fixed point that is a secondary minimum rather than in the main one.

Also, keep in mind that the dimension of the functional space is infinite and can have many unobvious secondary minima.

There can also be inflection and saddle points of the energy hypersurface (Figure 8 B). The trajectory completely stops at such a point.

It is a fundamental question whether or not such secondary fixed points as well as the inflection points belong to the problem under study. The answer is not straightforward. One should look for such an answer based on the origin of the equation.

Let us also mention possible gently sloping valleys in the energy relief. In this case, the motion along such a shallow slope may appear practically indistinguishable from an asymptotic falling into a fixed point during the numerical process.

■ 6. Summary

This article offers an approach to solve nonlinear stationary partial differential equations numerically. It is especially useful in the case of equations with zero boundary conditions that have both a trivial solution and nontrivial solutions. The approach is based on solving a pseudo-time-dependent equation instead of the stationary one, the initial condition being different from zero. Then the solver can avoid sticking to the trivial solution and is able to converge to a nontrivial solution. However, the penalty is increased simulation time.

■ References

- [1] M. M. Vainberg and V. A. Trenogin, *Theory of Branching of Solutions of Non-linear Equations*, Leyden, Netherlands: Noordhoff International Publishing, 1974.
- [2] E. M. Lifshitz and L. P. Pitaevskii, *Physical Kinetics: Course of Theoretical Physics*, Vol. 10, Oxford, UK: Pergamon, 1981 Chapter 101.
- [3] A. A. Bullbich and Yu. M. Gufan, "Phase Transitions in Domain Walls," *Ferroelectrics*, **98**(1), 1989 pp. 277–290. doi:10.1080/00150198908217589.
- [4] L. D. Landau and E. M. Lifshitz, *Quantum Mechanics: Course of Theoretical Physics*, Vol. 3, 3rd ed., Oxford, UK: Butterworth-Heinemann, 2003.
- [5] A. Boulbitch and A. L. Korzhenevskii, "Field-Theoretical Description of the Formation of a Crack Tip Process Zone," *European Physical Journal B*, **89**(261), 2016 pp. 1–18. doi:10.1140/epjb/e2016-70426-6.
- [6] A. Boulbitch, Yu. M. Gufan and A. L. Korzhenevskii, "Crack-Tip Process Zone as a Bifurcation Problem," *Physics Review E*, **96**(013005), 2017 pp. 1–19. doi:10.1103/PhysRevE.96.013005.

A. Boulbitch, "Pseudo-Dynamic Approach to the Numerical Solution of Nonlinear Stationary Partial Differential Equations," *The Mathematica Journal*, 2018. dx.doi.org/doi:10.3888/tmj.20-8.

About the Author

Alexei Boulbitch graduated from Rostov University (USSR) in 1980 and obtained his Ph.D. in theoretical solid-state physics in 1988 from this university. In 1990 he moved to the University of Picardie (France) and later to the Technical University of Munich (Germany). The Technical University of Munich granted him his habilitation degree in theoretical biophysics in 2001. His areas of interest are bacteria, biomembranes, cells, defects in crystals, phase transitions, physics of fracture (currently active), polymers and sensors (currently active). He presently works in industrial physics with a focus on sensors and gives lectures at the University of Luxembourg.

Alexei Boulbitch

Zum Waldeskühl 12

54298 Igel

Germany

abo@iee.lu